
ABSTRACT: This note introduces the Macaulay2 package SpaceCurves.m2 with
illustration. The 1.0 version of the package, provided in the accompanying online
supplement, is devoted to the generation of three types of curves in P3: smooth
curves, ACM curves and curves that are minimal in the even liaison class.

1. INTRODUCTION. The SpaceCurves project was initiated by R. Hartshorne,
F. Schreyer and M. Stillman during the 2017 Macaulay2 workshop at UC Berkeley.
Since the workshop, Zhang has improved old code and developed the package into
the present version. The goal of the SpaceCurves package is to generate three
types of curves in P3; smooth curves, ACM curves, and minimal curves in a given
even liaison class.

In Section 2 we illustrate how to produce smooth curves exhausting all possibil-
ities of (degree, genus) pairs. The philosophy is the following: first we construct
three types of surfaces; the smooth quadric surface, smooth cubic surfaces and
rational quartic surfaces with a double line. Next, we construct divisors on these
surfaces. Finally, we generate a random curve in a given divisor class.

In Section 3 we illustrate the stratification of the Hilbert scheme of ACM curves
in P3 using Betti tables. We illustrate how to produce ACM curves exhausting all
possibilities of Betti tables. First, we list all the possible Hilbert functions of ACM
curves of a given degree, then we produce all Betti tables of ACM curves with a
given Hilbert function. Finally, we generate a matrix of random forms with degrees
specified by the Hilbert–Burch degree matrix and take the ideal of maximal minors.

In Section 4 we explain the construction of a curve that is minimal in its even
liaison class from a given finite length module. The even liaison class can be
specified by either the ideal of a curve in the even liaison class, or by a finite length
module called the Hartshorne–Rao module. The implementation of the minimal
curve algorithm follows the paper by Guarrera et al. [1997].

To make our computations exact, and to avoid coefficient explosion, we work
over large finite fields Z/pZ in Macaulay2. Over a small prime field, such as Z/2Z

or Z/3Z, many of the curves produced will be singular. In these cases we refer the
readers to the package RandomCurvesOverVerySmallFiniteFields.m2 by C.
Bopp and F. Schreyer.

2. SMOOTH CURVES IN PROJECTIVE THREE-SPACE. For which pairs of integers
(d, g) does there exist a connected smooth curve in P3 of degree d and genus g?
G. Halphen gave partial answers in his prize winning treatise in 1882, and the com-
plete solution to this question was given by Gruson and Peskine almost a century
later in characteristic 0 and extended to characteristic p by Hartshorne.

(1) There are smooth plane curves of genus g = 1
2(d − 1)(d − 2) for any d ≥ 1.

(2) (Castelnuovo) If a smooth curve does not lie on any plane, we must have

g ≤
⌊ 1

4 d2
− d + 1

⌋
.

Any curve obtaining this bound lies on a quadric surface.

(3) For each a, b> 0, there are smooth curves on the smooth quadric surface with
degree d = a+ b and genus g = (a− 1)(b− 1).

(4) On the quadric cone, if d = 2a is even, there are smooth complete intersections
of the quadric cone with another surface of degree a. If d = 2a + 1 is odd,
then any degree d curve on the quadric cone has genus g = a2

− a.

(5) [Halphen 1882] If a curve does not lie on any plane or quadric surface, then

g ≤ 1
6 d(d − 3)+ 1.

(6) [Gruson and Peskine 1982, Corollary 2.3] For d ≥ 1 and
1
√

3
d3/2
− d + 1< g ≤ 1

6 d(d − 3)+ 1,

there is a smooth curve with degree d and genus g on a smooth cubic surface.

(7) [Gruson and Peskine 1982, Theorem 1.1; Hartshorne 1982, Theorem 0.2] For
d ≥ 1 and

0≤ g ≤ 1
8(d − 1)2,

there is a smooth curve with degree d and genus g on a smooth quartic surface
with a double line.

The SpaceCurves package generates curves on the surfaces mentioned above.
Let us start from curves on the smooth quadric surface.

2.1. Curves on a smooth quadric surface. To create a smooth quadric surface Q,
we use the method function quadricSurface(Ring) where the input ring is
taken as the ambient coordinate ring of P3. The output is a type of hashtable called
QuadricSurface. It contains the following information:

https://www.math.uni-sb.de/ag/schreyer/images/data/computeralgebra/M2/doc/Macaulay2/RandomCurvesOverVerySmallFiniteFields/html/index.html

i1 : needsPackage "SpaceCurves";
i2 : R = ZZ/101[x_0..x_3];
i3 : Q = quadricSurface(R)
o3 = ideal(- x x + x x )

1 2 0 3
o3 : QuadricSurface
i4 : peek Q
o4 = QuadricSurface{CanonicalClass => {-2, -2} }

HyperplaneClass => {1, 1}
Ideal => ideal(- x x + x x )

1 2 0 3
IntersectionPairing => | 0 1 |

| 1 0 |

Since Q ∼= P1
×P1, a basis of Pic(Q) consists of π∗1 (OP1(1)) and π∗2 (OP1(1)),

where π1 and π2 are the two canonical projections to P1. In this basis, the intersec-
tion pairing matrix is

[ 0
1

1
0

]
and the canonical class has coordinates {−2,−2}. The

divisor class {1, 1} in the given basis defines the embedding of Q in P3.
The method function divisor(List,QuadricSurface) produces divisors on

the quadric surface Q. The output is a type of hashtable called Divisor, which
carries a list that encodes the coordinates of the divisor as well as the surface it is
on. The intersection number is computed from the coordinates of D and E using
the intersection matrix of Q.

i5 : D = divisor({2,3},Q)
o5 = {2, 3}
o5 : Divisor
i6 : peek D
o6 = Divisor{Coordinate => {2, 3} }

Surface => ideal(- y*z + x*w)
i7 : E = divisor({1,2},Q);
i8 : D*E
o8 = 7

Since we can compute the intersection number of any two divisors on Q, we
can compute the degree and arithmetic genus of a divisor using Bezout’s theorem
and the adjunction formula:

deg D = D.H,

2pa(D)− 2= D.(D+ K ),

where H and K denote the hyperplane class and the canonical class, respectively.
The advantage of computing the degree and genus of a divisor abstractly is that the
coordinates of the divisor can be in any ring that supports rational arithmetic. In
particular, we can verify the formula for the degree and genus of a divisor of type
(a, b) on Q.

i9 : S = QQ[a,b];
i10 : F = divisor({a,b},Q);
i11 : degree F
o11 = a + b
o11 : S
i12 : genus F
o12 = a*b - a - b + 1
o12 : S

We use the method function curve(Divisor) to produce a random curve in
a given divisor class. The output type is called Curve, which is a hashtable that
encodes the ideal of the curve as well as the Divisor it comes from. To extract
the ideal, one could use the method ideal(Curve) or equivalently the key Ideal
of the hashtable Curve.

i13 : C = curve D;
i14 : I = ideal C;
i15 : (degree I, degree D, genus I, genus D)
o15 = (5, 5, 2, 2)
o15 : Sequence

Here is how the method curve(Divisor) works for divisors on a smooth
quadric surface. If D has coordinates {a, b}, we generate a random form f of
bidegree (a, b) in the Cox ring

Cox= k[s, t]⊗k k[u, v]

and create the Segre map ψ : R = k[x, y, z, w] → Cox /( f ) where

x 7→ s⊗ u, y 7→ s⊗ v, z 7→ t ⊗ u, w 7→ t ⊗ v.

The kernel of the map ψ will be the ideal of a curve in the divisor class (a, b) on
the quadric surface. Since we are using random bihomogeneous forms, the output
is typically smooth.

2.2. Curves on a smooth cubic surface. Although it is easy to produce a smooth
cubic surface, it is not straightforward to generate curves of a given degree and
genus on the given cubic surface. Instead, we realize the cubic surface as a blowup
π : X → P2 at six general points, anticanonically embedded into P3. Since the
automorphisms of P2 act transitively on four distinct points, we fix four points,

P1 = [1 : 0 : 0], P2 = [0 : 1 : 0], P3 = [0 : 0 : 1], P4 = [1 : 1 : 1],

and choose points P5 and P6 randomly. Then Pic(X)∼= Z7 with a basis given by
L ,−E1, . . . ,−E6, where L = π∗(OP2(1)) and Ei is the exceptional divisor of the

point Pi . The intersection matrix of X in this basis is given by[
1 0
0 − Id

]
,

where Id is the 6×6 identity matrix. The complete linear system |3L−E1−· · ·−E6|

is very ample and embeds X into P3 as a smooth cubic surface. The canonical class
of X is −3L + E1+ · · ·+ E6. See [Hartshorne 1977, V.4] for a treatment of these
well known facts.

We can use the method function cubicSurface(Ring) to generate a smooth
cubic surface. The output is a type of hashtable called CubicSurface. The key
BlowUpPoints stores the list of ideals of the six points. The key MapToP3 stores
the rational map from P2 to P3 which comes from the restriction of the embedding
X ↪→ P3.

i5 : X = cubicSurface(R)
o5 = ideal(...)
o5 : CubicSurface
i6 : peek X
o6 = CubicSurface{BlowUpPoints => {...} }

CanonicalClass => {-3, -1, -1, -1, -1, -1, -1}
HyperplaneClass => {3, 1, 1, 1, 1, 1, 1}
IntersectionPairing => | 1 0 0 0 0 0 0 |

| 0 -1 0 0 0 0 0 |
| 0 0 -1 0 0 0 0 |
| 0 0 0 -1 0 0 0 |
| 0 0 0 0 -1 0 0 |
| 0 0 0 0 0 -1 0 |
| 0 0 0 0 0 0 -1 |

Ideal => ideal(...)
MapToP3 => ...

The equation of X in P3 is computed in the following way: let I be the ideal of
the union of the six points P1, . . . , P6. Since P5 and P6 are chosen randomly, we
can presume that no three of Pi are collinear and that the points Pi do not all lie
on any quadric. The Hilbert–Burch theorem thus determines the shape of the free
resolution of I as

0→ R(−4)3 φ
→ R(−3)4→ I → 0,

where R = k[y0, y1, y2] is the coordinate ring of P2. Now φ is a 4× 3 matrix of
linear forms in three variables, which can be thought of as a 4×3×3 tensor. There
is a canonical way to consider φ as a 3× 3× 4 tensor, and think of it as a 3× 3
matrix of linear forms in four variables denoted by M ′. The determinant of M ′

gives the defining equation of X in P3.
We can create divisors on X using divisor(List,CubicSurface). Let us

compute the degree and genus of a divisor on the smooth cubic surface.

i17 : S = QQ[a,b_1..b_6];
i18 : D = divisor(gens S,X);
i19 : degree D
o19 = 3a - b - b - b - b - b - b

1 2 3 4 5 6
i20 : genus D
o20 = ...

2.3. Curves on a rational quartic surface. Let us recall the construction of the
quartic surface with a double line singularity from [Gruson and Peskine 1982].
Consider the blowup of P2 at nine general points P1, . . . , P9 given by π : Y → P2.
Let L denote the class π∗(O1

P(1)) on Y, and let Ei denote the class of the ex-
ceptional divisor corresponding to the point Pi . Then Pic(Y ) is isomorphic to
the free abelian group generated by L ,−E1, . . . ,−E9. There is a unique smooth
cubic curve 00 on P2 passing through the nine points, and let 0 denote its proper
transform on Y. We have 0= 3L−E1−· · ·−E9, and−0 is the canonical class of Y.
If we set C = L−E1, then the complete linear system |C+0| is basepoint-free and
maps Y into P3 as a quartic hypersurface. One can verify that the linear system
|C +0| separates points and tangent vectors for points in Y −0, and restricts to
a degree 2 linear system on the elliptic curve 0. As a result, Y − 0 is mapped
isomorphically onto its image, and 0 is mapped 2-1 to a line. The image of Y is a
rational quartic surface with a double line singularity, where the double line is the
scheme-theoretical image of the elliptic curve 0. For proofs of these statements,
see [Gruson and Peskine 1982, §1].

We use the method function quarticSurfaceRational(Ring) to create such
a surface. We can verify that the singular locus of the image of Y is indeed the
image of the elliptic 0 in Macaulay2.

i21 : Y = quarticSurfaceRational(R)
o21 = ideal(...)
o21 : QuarticSurfaceRational
i22 : peek Y
o22 = QuarticSurfaceRational{BlowUpPoints => {...

CanonicalClass => {-3, -1, -1, -1, -1, -1, -1, -1, -1, -1}
HyperplaneClass => {4, 2, 1, 1, 1, 1, 1, 1, 1, 1}
IntersectionPairing => | 1 0 0 0 0 0 0 0 0 0 |

| 0 -1 0 0 0 0 0 0 0 0 |
| 0 0 -1 0 0 0 0 0 0 0 |
| 0 0 0 -1 0 0 0 0 0 0 |
| 0 0 0 0 -1 0 0 0 0 0 |
| 0 0 0 0 0 -1 0 0 0 0 |
| 0 0 0 0 0 0 -1 0 0 0 |
| 0 0 0 0 0 0 0 -1 0 0 |
| 0 0 0 0 0 0 0 0 -1 0 |
| 0 0 0 0 0 0 0 0 0 -1 |

MapToP3 => ...
Ideal => ...

i26 : pts = intersect(Y.BlowUpPoints);
i31 : S = ring pts;
i27 : hilbertFunction(3,module pts)
o27 = 1
i28 : G = gens trim pts;
i32 : gamma0 = ideal (G*random(source G,S^{-3}));

-- This produces the unique plane cubic passing the 9 points.
o32 : Ideal of S
i36 : phi = map(S/gamma0,S);
i37 : psi = Y.MapToP3;
i39 : L = kernel (phi*psi)
o39 = ideal (y + 9918z - 2540w, x + 2540z - 11757w)
o39 : Ideal of R

-- This is the image of Gamma
i44 : radical ideal jacobian Y.Ideal
o44 = ideal (y + 9918z - 2540w, x + 2540z - 11757w)
o44 : Ideal of R

-- This is the singular locus of the image of Y
i45 : radical ideal jacobian Y.Ideal == L
o45 = true

The method function divisor(List,QuarticSurfaceRational) creates a
divisor on the quartic surface with a double line. We explain how the method
function curve(Divisor) turns Divisors into Curves on the CubicSurface
and the QuarticSurfaceRational. Given a divisor D = (a, b1, . . . , bn) where
n = 6 or 9, we generate a random plane curve C0 of degree a with multiplicity bi

at the point Pi and compute the equations of its image under the rational map
P2 99K P3.

Note that the produced curves have no components supported on the exceptional
curves.

2.4. Generating smooth curves. We say a divisor class is smooth if its general
members have smooth images in P3. The method function smoothDivisors gen-
erates all smooth divisors of a given degree on a given surface. So far the valid input
surfaces are QuadricSurface, CubicSurface and QuarticSurfaceRational.
On the smooth quadric surface as well as the smooth cubic surface, there are ex-
act numerical criteria to decide whether D is smooth [Hartshorne 1977, Exam-
ple V.4.8]. On the rational quartic surface however, we do not know of such a
criterion, and thus our list is not exhaustive. One sufficient condition for D to be
smooth is that D is basepoint-free and its restriction to 0 does not contain any pair
of involution points as basepoints.

The following code generates smooth divisors of degree 4 on all the supported
surfaces. Note that the output is a list of Divisors.

i2 : R = ZZ/32003[x_0..x_3];
i3 : Q = quadricSurface(R); X = cubicSurface(R);
i5 : Y = quarticSurfaceRational(R);
i6 : L = smoothDivisors(4,Q) | smoothDivisors(4,X) | smoothDivisors(4,Y);
i7 : netList L

+------------------------------+
o7 = |{1, 3} |

+------------------------------+
|{2, 2} |
+------------------------------+
|{2, 1, 1, 0, 0, 0, 0} |
+------------------------------+
|{3, 1, 1, 1, 1, 1, 0} |
+------------------------------+
|{2, 0, 1, 1, 1, 1, 0, 0, 0, 0}|
+------------------------------+
|{3, 1, 1, 1, 1, 1, 1, 1, 0, 0}|
+------------------------------+

Now we take the list of Divisors and turn them into Curves, and verify that
the curves we get are indeed smooth and have the correct degrees and genera.

i8 : LC = apply(L,D->curve D);
i9 : netList apply(LC,C-> (degree C,genus C, isSmooth C))

+------------+
o9 = |(4, 0, true)|

+------------+
|(4, 1, true)|
+------------+
|(4, 0, true)|
+------------+
|(4, 1, true)|
+------------+
|(4, 0, true)|
+------------+
|(4, 1, true)|
+------------+

The method function curve(ZZ,ZZ) takes a pair of integers (d, g), generates
divisors of degree d on all the surfaces, and selects one (if any) of genus g and
returns it as a Curve.

i11 : C = curve(8,5);
i12 : degree C, genus C
o12 = (8, 5)
o12 : Sequence

The following code generated 608 curves up to degree 15 in around three min-
utes on a desktop with quad cores Intel i5-7400 CPU at 3.00GHz.

i2 : R = ZZ/101[x_0..x_3];
i3 : X = quadricSurface(R);
i4 : Y = cubicSurface(R);
i5 : Z = quarticSurfaceRational(R);
i6 : dmax = 15;
i7 : time LD = flatten apply(splice{1..dmax},d->

smoothDivisors(d,X) | smoothDivisors(d,Y) | smoothDivisors(d,Z));
-- used 2.3459 seconds

i8 : time LC = apply(LD,D -> curve D);
-- used 185.549 seconds

i9 : #LC
o9 = 608

We use the method function dgTable to tally the number of curves by the degree
(horizontal axis) and genus (vertical axis).

i10 : dgTable LC
o10 = 42 | . . . . . . . . . . . . . 1

41 | . . . . . . . . . . . . . .
40 | . . . . . . . . . . . . . 1
39 | . . . . . . . . . . . . . .
38 | . . . . . . . . . . . . . .
37 | . . . . . . . . . . . . . .
36 | . . . . . . . . . . . . 1 1
35 | . . . . . . . . . . . . 1 .
34 | . . . . . . . . . . . . . .
33 | . . . . . . . . . . . . . .
32 | . . . . . . . . . . . . 1 .
31 | . . . . . . . . . . . . . 1
30 | . . . . . . . . . . . 1 . 2
29 | . . . . . . . . . . . . . 1
28 | . . . . . . . . . . . 1 . 1
27 | . . . . . . . . . . . . 1 3
26 | . . . . . . . . . . . . 1 1
25 | . . . . . . . . . . 1 . 1 3
24 | . . . . . . . . . . 1 1 2 3
23 | . . . . . . . . . . . . 1 3
22 | . . . . . . . . . . . 1 2 3
21 | . . . . . . . . . . 1 1 2 5
20 | . . . . . . . . . 1 . 2 4 3
19 | . . . . . . . . . . 1 1 2 3
18 | . . . . . . . . . 1 1 3 3 4
17 | . . . . . . . . . . 1 2 3 4
16 | . . . . . . . . 1 . 2 3 5 4
15 | . . . . . . . . 1 1 3 2 3 6
14 | . . . . . . . . . 2 1 3 4 3
13 | . . . . . . . . . 2 3 3 3 2
12 | . . . . . . . 1 2 1 3 4 4 5
11 | . . . . . . . . 1 2 3 2 3 2
10 | . . . . . . . 2 2 2 2 4 3 2

9 | . . . . . . 1 1 1 3 5 4 4 4
8 | . . . . . . 1 1 2 4 3 3 5 3
7 | . . . . . . 1 1 4 3 4 3 3 4
6 | . . . . . 1 2 5 5 3 6 5 6 6
5 | . . . . . 1 4 2 3 4 5 4 6 5
4 | . . . . 3 3 4 3 6 4 5 5 7 4
3 | . . . . 3 3 3 4 5 4 8 5 5 5
2 | . . . 3 3 2 5 3 5 4 6 3 7 6
1 | . 1 3 2 3 3 5 3 5 4 5 5 8 5
0 | 3 3 3 3 6 4 5 6 8 6 9 8 9 10

---+-------------------------------------------
g/d| 2 3 4 5 6 7 8 9 10 11 12 13 14 15

We close this section with a study of H sm
8,5 , the restricted Hilbert scheme of smooth

curves of degree 8 and genus 5 in P3. It is known that H sm
8,5 is irreducible; see for

example [Ein 1986, Theorem 4]. In [Gruson and Peskine 1978, §4], we find a
complete stratification of H sm

8,5 into locally closed families A, B, C, D, E, as well
as the Betti tables of each family. The following shows that the free resolutions for
the families E, C, B, A (in this order) are indeed correct.

i13 : L = select(LC, C -> degree C == 8 and genus C == 5);
i14 : L / ideal / res / betti

0 1 2 3 0 1 2 3 0 1 2 3 0 1 2 3
o14 = {total: 1 6 8 3, total: 1 5 5 1, total: 1 4 4 1, total: 1 7 8 2}

0: 1 . . . 0: 1 . . . 0: 1 . . . 0: 1 . . .
1: . 1 . . 1: . . . . 1: . . . . 1: . . . .
2: . . . . 2: . 1 . . 2: . 1 . . 2: . . . .
3: . . . . 3: . 3 3 . 3: . 3 2 . 3: . 7 8 2
4: . . . . 4: . 1 2 1 4: . . 2 1
5: . 5 8 3

o14 : List

We note that curves in family D are missing from our program because they do
not lie on any of the surfaces we constructed. In fact, they all lie on the ruled cubic
surface with a double line. In the upcoming 2.0 version of the package, we will
include curves on this surface and much more.

3. ACM CURVES IN PROJECTIVE THREE-SPACE. In this section we consider
arithmetically Cohen–Macaulay (ACM) curves in P3. The literature on ACM
curves in P3 is very rich and we do not aim to give a survey in this article. We
shall illustrate the stratification of Hilbert schemes of ACM curves by Betti tables.

3.1. Hilbert functions of ACM curves. Let H CM denote the points of the Hilbert
scheme (without fixing a Hilbert polynomial) of curves in P3 that correspond to
ACM curves. Ellingsrud [1975] shows that H CM is an open smooth subscheme of
the Hilbert scheme. Let H CM

H denote the points that correspond to ACM curves

with Hilbert function H , then we have a stratification

H CM
=

⊔
H

H CM
H .

Furthermore, it is shown in [Ellingsrud 1975] that the spaces H CM
H form disjoint

irreducible connected components of H CM.
What are all the possible Hilbert functions of ACM curves in P3?
There are many notions that encode information which is equivalent to that en-

coded by the Hilbert function, e.g., numerical characters as in [Gruson and Peskine
1978, §2], postulation characters as in [Martin-Deschamps and Perrin 1990, I.2]
and h-vectors as in [Migliore 1998, §1.4]. For computational reasons that we will
explain later, we will use the postulation character of a curve C, which is defined
to be the negative of the third discrete difference of its Hilbert function.

A theorem of Gruson and Peskine [1978, §2] says that the postulation charac-
ters of ACM curves are exactly the positive characters. A positive character is a
function γ : Z→ Z of finite support, such that the following hold:

(1)
∑

n γ (n)= 0,

(2) γ (n)= 0 for n < 0,

(3) γ (0)=−1,

(4) If we set s = s(γ )= inf{n|γ (n) 6= −1}, then γ (s)≥ 0,

(5) γ (n)≥ 0 for n ≥ s.

The degree of a positive character γ is defined to be
∑

n nγ (n). If γ is the
postulation character of a curve C , then the degree of γ is the degree of C , and
s(γ ) is the least degree surface C lies on.

Given d and s, the enumeration of all positive characters becomes a familiar
problem: what are the different ways to make d −

∑
n<s n cents in total using

s coins where each coin can have value n for every n ≥ s? Macaulay2 solves
this efficiently by creating a bigraded ring and enumerating monomials of a fixed
bidegree. The method function positiveChars(ZZ,ZZ) does exactly that, and
enumerates all positive characters with a given degree d and s. By looping s from 1
to d − 1, the method function positiveChars(ZZ) then enumerates all positive
characters of a given degree d .

i2 : positiveChars(6,2)
o2 = {{-1, -1, 1, 0, 0, 1}, {-1, -1, 0, 1, 1}}
o2 : List
i3 : positiveChars(6)
o3 = {{-1, -1, 1, 0, 0, 1}, {-1, -1, 0, 1, 1}, {-1, -1, -1, 3}}
o3 : List

We can produce all 121 positive characters up to degree 16 within 0.05 seconds.
The efficiency is one of the main reasons why we chose the postulation character

to represent the Hilbert function.

i4 : time L = flatten apply(15, d -> positiveChars(d+1));
-- used 0.0492106 seconds

i5 : #L
o5 = 121

3.2. Betti tables of ACM curves. We define a further stratification

H CM
H =

⊔
B

H CM
H,B,

where H CM
H,B consists of points that correspond to ACM curves with Hilbert func-

tion H and Betti table B.
What are all the Betti tables B of an ACM curve in P3?
A necessary and sufficient condition for a Betti table B of a homogeneous ideal

to be that of an ACM curve in P3 is the following: B has n + 1 generators of
degrees a1 ≥ · · · ≥ an+1 and n syzygies of degrees b1 ≥ · · · ≥ bn and no higher
syzygies, such that the diagonal entries of the matrix M = (ai − b j )i, j are positive.
See, for example, [Eisenbud 2005, Propositions 3.8 and 3.14]. We call M the
Hilbert–Burch degree matrix.

For a given Hilbert function H , let BH denote the set of Betti tables B such that
H CM

H,B is nonempty. We partially order BH by the number of generators (equiv-
alently syzygies) and write B >n B ′ if B has n more generators than B ′. Since
the Hilbert function H is fixed, the fourth discrete difference 14 H is equal to the
alternating sum of the Betti numbers

∑
i (−1)i Bi,n for every B ∈BH . Therefore

B >1 B ′ if and only if B is obtained from B ′ by adding one generator and one
syzygy of the same degree, and B >n B ′ if and only if B can be obtained from B ′

by n successive additions of one generator and one syzygy of the same degree. By
the previous paragraph, if we remove a generator and a syzygy of the same degree
from some B ∈BH to obtain B ′, then B ′ ∈BH also. Thus B is minimal in BH if
and only if no generator and syzygy of B share the same degree, but in this case B
is determined uniquely by 14 H =

∑
i (−1)i Bi,n . In conclusion: BH has a smallest

element denoted by b, and every B ∈BH can be obtained from b by successively
adding one generator and one syzygy of the same degree (although not all ways of
doing so yield a Betti table in BH ).

The main theorem of [Ellingsrud 1975] states that the downward closed strata⊔
B≤B ′ H

CM
H,B is open and irreducible (hence smooth) in H CM

H . Further, for two Betti
tables B, B ′ ∈BH , we have B ≥ B ′ if and only if H CM

H,B ′ ⊃H CM
H,B . In other words,

the partial order on BH corresponds to the partial order of specialization among
the strata H CM

H,B .
The method function generalACMBetti(List) takes a postulation character γ

corresponding to the Hilbert function H and returns the smallest element b of BH .

i2 : generalACMBetti {-1,-1,-1,2,1}
0 1 2

o2 = total: 1 3 2
0: 1 . .
1: . . .
2: . 3 1
3: . . 1

o2 : BettiTally

The method function specializeACMBetti(BettiTally) takes a Betti table
B ∈ BH and returns the list of all B ′ ∈BH such that B ′ >1 B.

i3 : specializeACMBetti oo
0 1 2

o3 = {total: 1 4 3}
0: 1 . .
1: . . .
2: . 3 2
3: . 1 1

o3 : List

Applying this iteratively on b, we exhaust all of BH in layers. The method
function allACMBetti(List) takes a postulation character, and returns all Betti
tables of ACM curves having that character. The following are all the Betti tables
of ACM curves of degree 8. Each row corresponds to a distinct Hilbert function,
and along each row the Betti tables are more special as we go from left to right.
Note that these outputs are only Betti tables, no curves are produced yet.

i5 : netList (positiveChars(6) / allACMBetti)
+------------+------------+
| 0 1 2| |

o5 = |total: 1 3 2| |
| 0: 1 . .| |
| 1: . 2 1| |
| 2: . . .| |
| 3: . . .| |
| 4: . 1 1| |
+------------+------------+
| 0 1 2| 0 1 2|
|total: 1 2 1|total: 1 3 2|
| 0: 1 . .| 0: 1 . .|
| 1: . 1 .| 1: . 1 .|
| 2: . 1 .| 2: . 1 1|
| 3: . . 1| 3: . 1 1|
+------------+------------+
| 0 1 2| |
|total: 1 4 3| |
| 0: 1 . .| |
| 1: . . .| |
| 2: . 4 3| |
+------------+------------+

The method function degreeMatrix(BettiTally) takes the Betti table of an
ACM curve, and returns its Hilbert–Burch degree matrix. The method function
randomDeterminantalIdeal(Ring,Matrix) computes the ideal of maximal
minors of a matrix of random forms with specified degrees. Together, these method
functions allow us to generate ACM curves. The following code generates an ACM
curve for each of the Betti table given above, and shows that they do indeed have
the predicted Betti tables.

i7 : L = positiveChars(6) / allACMBetti;
i8 : netList apply(L, H -> H / (B ->

betti res randomDeterminantalIdeal(ZZ/101[x,y,z,w],degreeMatrix B)))
+------------+------------+
| 0 1 2| |

o8 = |total: 1 3 2| |
| 0: 1 . .| |
| 1: . 2 1| |
| 2: . . .| |
| 3: . . .| |
| 4: . 1 1| |
+------------+------------+
| 0 1 2| 0 1 2|
|total: 1 2 1|total: 1 3 2|
| 0: 1 . .| 0: 1 . .|
| 1: . 1 .| 1: . 1 .|
| 2: . 1 .| 2: . 1 1|
| 3: . . 1| 3: . 1 1|
+------------+------------+
| 0 1 2| |
|total: 1 4 3| |
| 0: 1 . .| |
| 1: . . .| |
| 2: . 4 3| |
+------------+------------+

Here is an example of two layers of specializations. The second and the third
families are incomparable, and they both specialize to the fourth family.

i9 : gamma = (positiveChars(9))#5
o9 = {-1, -1, -1, 1, 1, 1}
o9 : List
i10 : allACMBetti gamma

0 1 2 0 1 2 0 1 2 0 1 2
o10 = {total: 1 2 1, total: 1 3 2, total: 1 3 2, total: 1 4 3}

0: 1 . . 0: 1 . . 0: 1 . . 0: 1 . .
1: . . . 1: . . . 1: . . . 1: . . .
2: . 2 . 2: . 2 1 2: . 2 . 2: . 2 1
3: . . . 3: . 1 . 3: . . 1 3: . 1 1
4: . . 1 4: . . 1 4: . 1 1 4: . 1 1

o10 : List

We end this section with a short comment on smoothness. When does H CM
H,B

have a point that corresponds to a smooth ACM curve? The answer is given by
[Geramita and Migliore 1989, Proposition 1.3]: if and only if the Hilbert–Burch
degree matrix has positive upper diagonal. We leave it as an exercise to the reader
to check which of the four families above have a smooth ACM curve.

4. MINIMAL CURVES IN AN EVEN LIAISON CLASS. The Hartshorne–Rao mod-
ule (or deficiency module) of a curve C is defined to be the graded module

MC := H 1
∗
(IC)=

⊕
n∈Z

H 0(IC(n))

over the polynomial ring S = H 0
∗
(OP3). Since the curves we consider are locally

Cohen–Macaulay and equidimensional, one can show that MC is of finite length.
The method raoModule(Ideal) computes the Hartshorne–Rao module from

the ideal of a curve using local duality. We compute a free resolution F• of S/IC ,
then dualize and take the cokernel of the last term to compute Ext3S(S/IC , S). Then
we resolve Ext3S(S/IC , S) by G• and again dualize and take the cokernel of the last
term to obtain MC . The following is an example of the rational quartic curve.

ii78 : I = monomialCurveIdeal(R,{1,3,4});
oo78 : Ideal of R
ii79 : betti res I

0 1 2 3
oo79 = total: 1 4 4 1

0: 1 . . .
1: . 1 . .
2: . 3 4 1

oo79 : BettiTally
ii80 : M = raoModule I
oo80 = cokernel {1} | w -z y -x |

1
oo80 : R-module, quotient of R
ii81 : betti res M

0 1 2 3 4
oo81 = total: 1 4 6 4 1

1: 1 4 6 4 1
oo81 : BettiTally

Rao [1978/79] made the beautiful discovery that there is a bijection between the
even liaison classes of curves and isomorphism classes of finite length modules up
to a shift in grading. We say a curve C is minimal if for every curve D in the even
liaison class of C , we have MD ∼=MC [h] for some h≥ 0. It turns out that a minimal
curve also has the minimal degree and genus among all curves in its even liaison
class, and they are unique up to deformation with constant cohomology. Further-
more, all curves in the even liaison class can be obtained from C by finitely many

basic double links and a deformation with constant cohomology. In particular, the
possible degrees and genera of all curves in the even liaison class can be computed
from the minimal curve C alone. This vastly generalizes the case of ACM curves
where H 1

∗
(IC)= 0. We refer the readers to [Martin-Deschamps and Perrin 1990]

for details.
Given a finite length graded S-module M, we wish to construct a minimal

curve C such that MC ∼= M[h] for some smallest possible integer h. A suitable
algorithm is outlined in [Martin-Deschamps and Perrin 1990] and improved by
[Guarrera et al. 1997]. We give a simplified account here. Consider the short exact
sequence of S-modules

0→ S/IC → H 0
∗
(IC)→ MC → 0.

If F• and G• are minimal free resolutions of S/IC and H 0
∗
(IC), respectively, and

α : F•→G• is an induced map, then the mapping cone C(α)• is a resolution of MC .
By [Martin-Deschamps and Perrin 1990, Theorems 3.7 and 4.1], if C is a minimal
curve, then C(α)• is actually minimal. Since H 0

∗
(IC) has depth 2, it has projective

dimension 2 by the Auslander–Buchsbaum formula. We see that F2 ∼= L4, F1 ∼= L3

and F0 is a summand of L2.
We turn this observation around, and start with a minimal free resolution L• of

a finite length module M and ask: what are the conditions on the projections π
such that coker d = coker(π ◦ d3) is the ideal of a curve in P3 after shifting by an
integer h?

The Buchsbaum–Eisenbud criterion of exactness tells us that if the lower com-
plex in

0 // L4 // L3
d3
// L2

π

��

// �2 M

��

// 0

0 // L4 // L3
d
// P // coker d // 0

is a free resolution of an ideal (up to shift), then rk P = rk L3 + 1 − rk L4 =

rk d+1 and grade I (d)≥ 2. It turns out that this is also sufficient. We rephrase the
grade I (d)≥ 2 condition in terms of the ideal of maximal nonvanishing minors: this
is if and only if I (d) is not contained in any principal ideal, i.e., rk d⊗S S/( f )= rk d
for all f ∈ S. Since the alternating sum of the degrees of the free modules must
be zero for the free resolution of a curve by the Herzog–Kühl equations, we can
easily determine h whenever the degrees of P are known. To summarize, we are
looking for projections π satisfying the above conditions such that the resulting
shift h is as small as possible.

The key observation of [Guarrera et al. 1997] is that the rank and the rank in codi-
mension 1 of a matrix M over a PID can be easily read off from its Smith normal
form. Furthermore, if we evaluate the matrix d in a PID k[T ], then for a general

choice of evaluations k[x0, . . . , x3] → k[t], the rank and rank in codimension 1 of
d are preserved. The method function minimalCurve(Module) implements this
method to find a projection π corresponding to a minimal curve.

Let us generate a minimal curve in the even liaison class of the rational quartic.

ii82 : J = minimalCurve M;
oo82 : Ideal of R
ii83 : betti res J

0 1 2 3
oo83 = total: 1 4 4 1

0: 1 . . .
1: . 4 4 1

oo83 : BettiTally

Indeed, the rational quartic curve has Hartshorne–Rao module k concentrated
in degree 1, and lies on a smooth quadric with three other cubic generators. The
complete intersection of the quadric with a general linear combination of the cubic
generators is a divisor of type (3, 3) on the quadric surface. Since the rational
quartic is of type (1, 3), the residual would have type (2, 0)— the union of two
skew lines. The latter has Hartshorne–Rao module k concentrated in degree 0, and
is obviously of minimal degree in its even liaison class.

Finally, here is a remark about the function minimalCurve(Module). Once
the matrix d is computed, there are two ways to obtain the equations of the ideal
IC : the first is to compute the kernel of the transposition of d , the second is to take
maximal minors and saturate by the irrelevant ideal. Unfortunately, both would
take a long time if d is a large matrix. To get partial information, the method func-
tion minimalCurveBetti(Module) outputs the Betti table of a minimal curve
corresponding to a finite length module without explicitly computing the ideal of
the curve.

